
Tabeller
En viktig tillämpning är tabeller – att ifrån en ”nyckel” kunna ta fram 
ett ”tabellvärde”. Ett typiskt exempel är en telefonkatalog:

Avdelningen för datalogi
Andersson, Arne, 1022
Bol, Roland, 7606
Eriksson, Lars-Henrik, 1057
Flener, Pierre, 1028

           nyckel  värde       

Operationer på tabellen
Tabellen i exemplet kopplar ett heltalsvärde till en sträng.
Vad vill man kunna göra för beräkningar med sådana tabeller?
• Lägg in en uppgift i en tabell (update)
• Hämta ett värde från en tabell (lookup)
• Tag bort en uppgift från en tabell (remove)
(Fler operationer är tänkbara, vi kommer att diskutera några till, men 
nöjer oss med dessa just nu.)

Not: Man skriver lite slarvigt att man ”lägger in” eller ”tar bort” 
uppgifter, men tabeller representeras förstås av datastrukturer som 
all annan information. update och remove beräknar alltså nya 
tabeller med förändrat innehåll, snarare än ändrar den gamla.

Specifikationer för tabellfunktionerna
Vi antar att tabellernas datatyp har namnet table.

(* lookup(key,table)
   TYPE: string*table -> int option
   PRE:  table är en korrekt tabell.
   POST: SOME v om key finns i table och v är
         motsvarande värde. NONE annars. *)
(* update(key,value,table)
   TYPE: string*int*table -> table
   PRE:  table är en korrekt tabell.
   POST: Tabellen table uppdaterad med värdet
         value för nyckeln key *)
(* remove(key,table)
   TYPE: string*table -> table
   PRE: table är en korrekt tabell.
   POST: Tabellen table med uppgiften om nyckeln
         key borttagen *)

Den tomma tabellen
Innan man kan börja arbeta med tabeller så måste man ha en första 
tabell att utgå ifrån. Låt oss anta att det finns ett namngivet värde 
empty som är en tom tabell.

Användning av tabellen
Vi låter någon programmera tabellfunktionerna åt oss och sedan 
kan vi skriva program som använder tabeller!
val datalogi =
  update("Andersson, Arne",1022,
   update("Bol, Roland",7606,
    update("Eriksson, Lars-Henrik",1057,
     update("Flener, Pierre",1028,empty))));

lookup("Bol, Roland",datalogi) ––> SOME 7606
lookup("Anka, Kalle",datalogi) ––> NONE
lookup("Bol, Roland",
 update("Bol, Roland",1234,datalogi)) ––> SOME 1234
lookup("Bol, Roland",
 remove("Bol, Roland",datalogi)) ––> NONE
Det verkar fungera bra, men fattas inte något...?

Abstrakta datatyper
Vi har inte sagt någonting om hur tabellerna skall representeras. 
Tydligen kunde vi använda dem ändå!

Kunskapen om hur tabeller ser ut finns i koden för funktionerna 
lookup, update och remove. Den som skriver dessa funktioner 
måste veta hur tabeller representeras, men inte den som använder 
tabellerna.

Användningen av tabeller är alltså oberoende av representationen.  
Man säger att representationen av tabellen har abstraherats bort 
genom dataabstraktion och att tabellen är en abstrakt datatyp.

De tre funktionerna och värdet empty utgör tabelldatatypens 
gränsyta (eng. interface) mot den kod som använder tabeller.

Implementering av tabellen
Ett sätt att implementera tabellen är som en lista.
type table = (string*int) list;

datalogi ––> 
[("Andersson, Arne",1022),
 ("Bol, Roland",7606),
 ("Eriksson, Lars-Henrik",1057),
 ("Flener, Pierre",1028)] : table;
            
En invariant är att varje nyckel förekommer högst en gång – i annat 
fall skulle tabellvärdet inte vara entydigt bestämt.

Tabellprogrammet – lookup
(* lookup(key,table)
   TYPE: string*table -> int option
   PRE: table är en korrekt tabell.
   POST: SOME v om key finns i table och v är
         motsvarande värde. NONE annars. *)
(* VARIANT: längden hos table *)
fun lookup(_,[]) = NONE
  | lookup(key,(key',value)::rest) =
      if key = key' then
        SOME value
      else
        lookup(key,rest);

Tabellprogrammet  – empty, update
val empty = [];

(* update(key,value,table)
   TYPE: string*table -> table
   PRE:  table är en korrekt tabell.
   POST: Tabellen table uppdaterad med värdet
         value för nyckeln key *)
(* VARIANT: längden av table *)
fun update(key, value, []) = [(key,value)]
  | update(key, value, (key',value')::rest) =
      if key = key' then
        (key,value)::rest
      else
        (key',value')::update(key,value,rest);

Observera att update respekterar datastrukturinvarianten – att 
varje nyckel förekommer högst en gång i tabellen.

Tabellprogrammet  – remove
(* remove(key,table)
   TYPE: string*table -> table
   PRE: table är en korrekt tabell.
   POST: Tabellen table med uppgiften om nyckeln
         key borttagen *)
(* VARIANT: längden av table *)
fun remove(key, []) = []
  | remove(key, (key',value)::rest) =
      if key = key' then
        rest
      else
        (key',value)::remove(key,rest);

Observera att även remove respekterar datastrukturinvarianten.

Tabelltyper
Vi har i specifikationerna skrivit att nycklarna skall vara strängar och 
tabellvärdena heltal, men tabellfunktionerna blir faktiskt polymorfa:
lookup : ''a*(''a*'b) list -> 'b option
update : ''a*'b*(''a*'b) list -> (''a*'b) list
remove : ''a*(''a*'b) list -> (''a*'b) list

Anledningen är att programmet aldrig utför någon operation på 
nycklar eller värden (bara likhetsjämförelser på nycklar) och därför 
inte bryr sig om vilka typer dessa har.

Funktionerna arbetar med tabeller på formen (''a*'b) list.

Vår tabell hade typen (string*int) list, vilket är ett speciallfall 
– eller instans – av den polymorfa tabelltypen ovan.

Stöd för dataabstraktion i ML
För att förhindra att användaren av en abstrakt datatyp av misstag 
(eller avsiktligt...) gör sig beroende av representationen så har ML 
stöd för att tvinga användningen att bli representationsoberoende. 
abstype-deklarationen är en variant av datatype-deklarationen.

abstype datatypdeklaration
  with
    deklarationer av gränsytan
  end;

Operationer i den abstrakta datatypens gränsyta definieras mellan 
with och end. Konstruktorerna i datatypsdeklarationen blir osynliga 
och oåtkomliga utom just mellan with och end. Detta tvingar 
datatypen att bli abstrakt.
Av samma skäl skrivs inte abstype-värden ut av ML-systemet.

Tabellen som abstype
För att göra tabellen till en abstype så kan man definiera en 
konstruktor Table och tagga listorna i tabellen med den.
abstype table = Table of (string*int) list
with
  val empty = Table [];
  fun lookup(key,Table table) = ...;
  fun update(key,Table table) = ...;
  fun remove(key,Table table) = ...;
end;

Olika sorters operationer i gränsytan
Man brukar skilja på
• Operationer som konstruerar nya värden av den abstrakta 
  datatypen (t.ex. update). De kallas ofta konstruktorer. (Men 
  blanda inte ihop dem med konstruktorer som används för att tagga
  data – t.ex. Table).
• Operationer som hämtar delar av värden (t.ex. lookup). De kallas
   ofta selektorer.
• Operationer som gör tester eller jämförelser på värden av den 
   abstrakta datatypen. De kallas ofta predikat.
Gränserna mellan dessa är dock inte skarp.

Intern dokumentation av abstrakta datatyper
Varje abstype skall ha intern dokumentation på samma sätt som 
en datatype.

Dessutom skall alla definitioner av värden (val) som ingår i 
gränsytan också dokumenteras med
• Namn på värdet
• Värdets typ
• Beskrivning av vad värdet representerar

T.ex.:
  (* empty
     TYPE: table
     VALUE: En tom tabell *)
  val empty = Table [];

Fördelar med dataabstraktion
• Operationerna kan implementeras oberoende av användningen av
   den abstrakta datatypen. Den som skall använda den abstrakta 
   datatypen behöver inte veta hur datatypen är implementerad.
• Eftersom den som använder den abstrakta datatypen inte heller 
  får utnyttja eventuell kunskap om hur datatypen är implementerad 
  så kan man ändra implementeringen utan att behöva ändra 
  någonting i övriga delar av programmet. (Så länge som gränsytans
  funktionsspecifikationer följs!)
• Av samma skäl så behöver den som skall använda datatypen inte
  ta något ansvar för invarianter – det vilar helt på implementeringen. 
• Den tydliga gränsen mellan implementering och användning gör
   det lättare att återanvända implementeringen i andra situationer.

Nackdelar med dataabstraktion
• Man kan bara göra sådana operationer som är förutsedda.
  Exempel: Om man vill veta om tabellen är tom, så går inte det!
  (En nödlösning kan vara att lägga till en operation i gränsytan som
  ”exporterar” all information i datatypen i något bestämd format och 
  sedan skriva en egen funktion som gör arbetet.)
• Man kan förlora prestanda i vissa fall.
  Exempel: Om man vet att en nyckel inte finns i tabellen så kan
  man lägga in en ny uppgift för nyckeln i konstant tid genom att
  skriva (x,y)::table i stället för update(x,y,table).
• Speciellt för ML: Eftersom konstruktorerna inte går att använda 
  utanför abstype-deklarationen så kan de inte används i 
  matchning utanför. Det kan ge mer svårlästa program.

Likhetstyper
Abstrakta datatyper i ML är inte likhetstyper.
Värden med samma ”beteende” kan ha olika representation!
Exempel:
val tab1 = update("A",1,update("B",2,empty));
val tab2 = update("B",2,update("A",1,empty));

tab1 och tab2 är nu bundna till tabeller med samma information.
Men tab1≠tab2, därför att listorna har elementen i olika ordning.
För att förhindra misstag och för att inte fakta om 
datarepresentationen skall avslöjas tillåter inte ML jämförelser 
mellan värden av abstrakta datatyper (utanför  with ... end).

För att kunna jämföra måste gränsytan ha en operation för 
jämförelsen som då kan ta hänsyn till representationen.
PK 2009/10 moment 8 Sida 19 Uppdaterad 2010-01-18

Sammanfattning av abstraktion
Definitionsabstraktion – ett värde ersätts av ett symboliskt namn
   Ex.: x < maximum  i stället för  x < 100
(vitsen är bättre läsbarhet och att behöver man ändra maximum 
räcker det med att ändra på ett ställe – där maximum definieras.)
Funktionsabstraktion – ett uttryck görs oberoende av specifika data
   Ex.: (fn x => x+1)y  i stället för  y+1
(vitsen är att (fn x => x+1) kan namnges och användas i många 
olika sammanhang – ofta även bättre läsbarhet.)
Dataabstraktion – ett program görs oberoende av specifik
datarepresentation
   Ex.: update(x,y,table)  i stället för  (x,y)::table
(vitsen är att man kan förändra representationen utan att ändra 
programmet, lättare att uppfylla invarianter, ofta bättre läsbarhet.)

Tabellkomplexitet
Vi har representerat en tabell med n nycklar av en lista med n 
element.

För att slå i tabellen måste man i genomsnitt söka igenom halva 
listan – dvs n/2 rekursiva anrop av lookup.

En lista med en miljon nycklar kräver i genomsnitt 500 000 anrop av 
lookup för att göra en tabellslagning. Detta är inte bra i praktiken.

Tidsåtgången för lookup är alltså proportionell mot antalet nycklar i 
tabellen. Komplexiteten hos lookup är O(n). Detsamma gäller 
update och remove.

Vi skall se på en annan implementering som ger lägre komplexitet.

Trädrepresentation
Tabellinformation kan lagras i träd i stället för i listor.

4

2

1 6 9

7

Varje nod innehåller en nyckel och motsvarande värde.
Detta träd representerar en tabell med nycklarna "1", "2", "4", 
"6", "7" och "9". (Tänk på att nycklarna är strängar!)
I exemplen kommer tabellvärdena att vara heltalet som motsvarar 
nyckeln. Av utrymmesskäl visas bara nycklarna i figurerna.

Binära sökträd
Ett binärt träd kallas för ett binärt sökträd om den har invarianten:
• Alla noder i vänster delträd har nycklar med lägre värden än
  rotnodens nyckel.
• Alla noder i höger delträd har nycklar med högre värden än
  rotnodens nyckel
• Dessa villkor gäller även för varje delträd

4

2

1 6 9

7

Sökning i binära sökträd
Sök efter nyckeln "6". Starta vid roten.
Vid varje nod väljer man vänster eller höger sökträd beroende på 
om det man söker efter är mindre eller större än nodens nyckel.

4

2

1 6 9

7

större

mindre

träff

Tidkomplexitet för binära sökträd
Om alla ”nivåer” i trädet är fyllda så innehåller ett träd med n nivåer 
2n-1 noder. För att söka efter en nyckel behöver man bara besöka 
så många noder som motsvarar antalet nivåer.

Antalet rekursiva anrop för ett träd 2n-1 noder är alltså n. 
Tidsåtgången är proportionell mot logaritmen av antalet nycklar – 
O(log n).

Ett träd med en miljon nycklar kräver c:a 20 rekursiva anrop av 
lookup – jämför med 500 000 som krävdes när tabellen 
representerades som en lista.

Detta förutsätter att trädet är ”välbalanserat” – mera senare...

Implementering av tabeller
som binära sökträd

Vi kan nu göra en alternativ implementering av tabeller som binära 
sökträd. Gränsytan med sina funktionsspecifikationer är helt 
oförändrad.

Tack vara att datatypen är abstrakt så kommer inga program som 
använder tabelldatatypen att påverkas av ändringen.

Man kan använda tabellen på exakt samma sätt som när den var 
implementerad som en lista.

Deklaration av tabeller som binära sökträd
abstype table =
   Empty
 | Bintree of string*int*table*table
(* REPRESENTATION CONVENTION:
     Ett tomt träd representeras av Empty. Varje icketomt 
     träd representeras av Bintree(k,v,l,r), där k är 
     nyckeln, v är tabellvärdet, l är vänster delträd och r 
     är höger delträd. 
   REPRESENTATION INVARIANT:
     Alla nycklar i vänster resp. höger delträd 
     måste vara mindre resp. större än k  *)
with
  (* empty
     TYPE: table
     VALUE: Ett tomt träd *)
  val empty = Empty;
  fun update .....;
  fun lookup .....;
  fun remove .....;
end;

Funktionen lookup
(* lookup(key,table)
   TYPE: string*table -> int option
   PRE:  table är en korrekt tabell.
   POST: SOME v om key finns i table och v är
         motsvarande värde. NONE annars. *)
(* VARIANT: Största djupet hos ett löv i table *)
  fun lookup(key,Empty) = NONE
    | lookup(key,Bintree(key',value,left,right)) =
        if key = key' then
          SOME value
        else if key < key' then
          lookup(key,left)
        else
          lookup(key,right);

Körexempel
lookup("6",Bintree("4",4,
               Bintree("2",2,
                    Bintree("1",1,Empty,Empty),
                    Empty),
               Bintree("7",7,
                    Bintree("6",6,Empty,Empty),
                    Bintree("9",9,Empty,Empty))))
––>  if "6" = "4" then
      SOME 4
    else if "6" < "4" then
      lookup("6",Bintree("2",2,
                     Bintree("1",1,Empty,Empty),
                     Empty))
    else
      lookup("6",Bintree("7",7,
                     Bintree("6",6,Empty,Empty),
                     Bintree("9",9,Empty,Empty)))

Körexempel (forts.)
––>  lookup("6",Bintree("7",7,
                   Bintree("6",6,Empty,Empty),
                   Bintree("9",9,Empty,Empty)))
––>  if "6" = "7" then
      SOME 7
    else if "6" < "7" then
      lookup("6",Bintree("6",6,Empty,Empty))
    else
      lookup("6",Bintree("9",9,Empty,Empty))
––>  if "6" = "6" then
      SOME 6
    else if "6" < "6" then
      lookup("6",Empty)
    else
      lookup("6",Empty)
––>  SOME 6

Uppdatering av binära sökträd
För att sätta in nyckeln "3": Sök efter den och sätt in den där den 
borde ha funnits:

1 6 9

mindre

hittar inte

ny nod

3

Funktionen update
(* update(key,value,table)
   TYPE: string*int*table -> table
   PRE:  table är en korrekt tabell.
   POST: Tabellen table uppdaterad med värdet
         value för nyckeln key *)
(* VARIANT: största djupet av ett löv i table *)
  fun update(key,value,Empty) =
        Bintree(key,value,Empty,Empty)
    | update(key,value,
             Bintree(key',value',left,right)) =
        if key = key' then
          Bintree(key,value,left,right)
        else if key < key' then
          Bintree(key',value',
                  update(key,value,left),right)
        else
          Bintree(key',value',
                  left,update(key,value,right));

Obalans
Sätter man i tur och ordning in "1", "2", "4", "6", "7" och "9" i 
ett tomt sökträd så får man:

1
2

4
6

7
9

Detta är ett kraftigt obalanserat träd. I praktiken en lista, vilket ger 
samma tidsåtgång för sökning som i en lista – linjärt i storleken.

För att binära sökträd skall vara praktiskt användbara måste de 
balanseras om efter uppdateringar.
Balanserade sökträd behandlas i nästa kurs (AD1).

Funktionen toList
Ibland vill man hämta ut informationen i ett träd som en lista. Vi 
utökar gränsytan med funktionen toList.
(* toList table
   TYPE: table -> (string*int) list
   PRE: (inget)
   POST: En lista av alla nyckel-värde-par i
         tabellen i inorder. *)
(* VARIANT: Största djupet av något löv i table *)
  fun toList Empty = []
    | toList(Bintree(key,value,left,right)) =
        toList left @ (key,value) :: toList right;

toList går igenom (traverserar) hela trädet för att samla 
information från noderna och bygga en lista. 
toList av exempelträdet ger 
[("1",1),("2",2),("4",4),("6",6),("7",7),("9",9)]

Mer om traversering
toList ”besöker” vänster delträd först, sedan rotnoden, sedan 
höger delträd – inorder-traversering.

Notera att listan som toList returnerar är sorterad. Genom att 
bygga ett sökträd med en mängd data och sedan göra om det till en 
lista i inorder så har man sorterat datamängden! (Ett alternativ till 
mergesort som kallas trädsortering.)

preorder-traversering – man besöker först rotnoden, sedan vänster 
delträd, sedan höger.

postorder-traversering – man besöker först vänster delträd, sedan 
höger, sedan rotnoden.

Polymorfa tabeller
Vår tabelldatatyp är definierad med strängnycklar och heltalsvärden. 
I praktiken vill man kunna definiera en abstrakt datatyp för tabeller 
som kan hantera alla typer av nycklar och värden eftersom en 
sådan är mera användbar.

Funktionsspecifikationerna behöver då ändras för att bli mera 
generella, t.ex.
(* lookup(key,table)
   TYPE: ''a*(''a*'b) table -> 'b option
   PRE: table är en korrekt tabell.
   POST: SOME v om key finns i table och v är
         motsvarande värde. NONE annars. *)

Notera att typvariabeln för nyckeltypen endast avser likhetstyper 
(dubbel apostrof). Varför?

Polymorfa tabeller (forts.)
Som vi såg tidigare så var funktionerna som implementerade 
tabeller som listor redan fullständigt polymorfa. Med deklarationen
abstype (''a,'b) table = Table of (''a*'b) list

så blir funktionernas typer de önskade
lookup : ''a*(''a*'b) table -> 'b option
update : ''a*'b*(''a*'b) table -> (''a*'b) table
remove : ''a*(''a*'b) table -> (''a*'b) table

Det är inte lika enkelt med implementeringen av tabeller som binära 
sökträd. Med en deklaration
abstype (''a,'b) table = Empty
 | Bintree of ''a*'b*(''a,'b) table*(''a,'b) table
så blir funktionerna ”av sig själva” polymorfa i tabellvärdet, men inte 
i nycklarna. Varför?

Operationer på sökträd
 blir inte helt polymorfa

I implementeringen av tabeller som listor användes 
likhetsjämförelse för att jämföra nycklar. = är en polymorf funktion.

I implementeringen av tabeller som binära sökträd användes 
storleksjämförelse av nycklar. Storleksjämförelser är beroende av 
typen: < är en monomorf funktion. Den är överlastad så den kan 
fungera med ett antal olika typer (string, int m.fl.), men bara en 
typ vid varje gång den förekommer i programkoden.

Detta händer även om själva sökträden är polymorfa!

Detta problem kan man lösa med högre ordningens funktioner 
(kursmoment 9).

